Sukzessives Aufblasen eines ausgearteten stationären Punkts

Ausgangssystem hat einen (reellen) stationären Punkt im Ursprung. Jacobi-Matrix verschwindet.

```
f_0 := matrix(2,1,[ y^2+x^7, y^3*(1+x^2)]);
sol := solve(f_0,[x,y]); sol := sol[1]:
print(sol,subs(jacobian(f_0,[x,y]),sol))
```

$$\begin{pmatrix} x^7+y^2 \\ y^3\,(x^2+1) \end{pmatrix}$$

$$\left\{[x=0,\, y=0],\, \left[x=-\mathrm{i},\, y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\, \left[x=-\mathrm{i},\, y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\, \left[x=\mathrm{i},\, y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\right.$$
$$\left.\left[x=\mathrm{i},\, y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right]\right\}$$

$$[x=0,\, y=0],\, \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Bestimmung des minimalen Grades im Vektorfeld und des charakteristischen Polynoms.

```
m := min(op(map(f_0,ldegree)));
fm := map(map(map(f_0,monomials), select, p -> degree(p)=m), _plus@op);
F := factor(x*fm[2]-y*fm[1])
```

$$2$$

$$\begin{pmatrix} y^2 \\ 0 \end{pmatrix}$$

$$-y^3$$

y=0 ist die einzige charakteristische Richtung und nicht einfach. Daher führen wir eine erste Aufblasung in x-Richtung durch. Der Ursprung ist weiterhin der einzige stationäre Punkt und die Jacobi-Matrix verschwindet immer noch.

```
Psi_x := [ xx, xx*yy ];
SPsi_x := [ x=xx, y=xx*yy ]:
J_x := jacobian(Psi_x, [xx,yy]):
JI_x := simplify(J_x^(-1)):
```

$$[xx,\, xx\, yy]$$

```
f_1 := subs(normal(JI_x*subs(f_0,SPsi_x)), [xx=x,yy=y]); f_1 := map(normal(f_1/x^(m-1)),factor);
sol := solve(f_1,[x,y]); sol := sol[1]:
print(sol,subs(jacobian(f_1,[x,y]),sol))
```

$$\begin{pmatrix} x^7+x^2\,y^2 \\ -x^6\,y+x^4\,y^3+x^2\,y^3-x\,y^3 \end{pmatrix}$$

$$\begin{pmatrix} x\,(x^5+y^2) \\ -y\,(x^5-x^3\,y^2-x\,y^2+y^2) \end{pmatrix}$$

$$\left\{[x=0,\, y=0],\, \left[x=-\mathrm{i},\, y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\, \left[x=-\mathrm{i},\, y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\, \left[x=\mathrm{i},\, y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\right.$$
$$\left.\left[x=\mathrm{i},\, y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right]\right\}$$

$$[x=0,\, y=0],\, \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Wieder Bestimmung des charakteristischen Polynoms.

```
m := min(op(map(f_1,ldegree)));
fm := map(map(map(f_1,monomials), select, p -> degree(p)=m), _plus@op);
F := factor(x*fm[2]-y*fm[1])
```

$$3$$

$$\begin{pmatrix} x\,y^2 \\ -y^3 \end{pmatrix}$$

$$-2\,x\,y^3$$

x=0 ist eine einfache charakteristische Richtung und invariant. y=0 ist nicht einfach als charakteristische Richtung, deshalb führen wir eine zweite x-Aufblasung durch. Nach wie vor gibt es nur den Ursprung als stationären Punkt mit der Nullmatrix als Jacobi-Matrix.

```
f_2 := subs(normal(JI_x*subs(f_1,SPsi_x)), [xx=x,yy=y]); f_2 := map(normal(f_2/x^(m-1)),factor);
sol := solve(f_2,[x,y]); sol := sol[1]:
print(sol,subs(jacobian(f_2,[x,y]),sol))
```

$$\begin{pmatrix} x\,(x^5+x^2\,y^2) \\ x^5\,y^3-2\,x^5\,y+x^3\,y^3-2\,x^2\,y^3 \end{pmatrix}$$

$$\begin{pmatrix} x\,(x^3+y^2) \\ y\,(x^3\,y^2-2\,x^3+x\,y^2-2\,y^2) \end{pmatrix}$$

$$\left\{[x=0,\,y=0],\,\left[x=-\mathrm{i},\,y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\,\left[x=-\mathrm{i},\,y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\,\left[x=\mathrm{i},\,y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\right.$$
$$\left.\left[x=\mathrm{i},\,y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right]\right\}$$

$$[x=0,\,y=0],\,\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Das charakteristische Polynom verändert sich kaum.

```
m := min(op(map(f_2,ldegree)));
fm := map(map(map(f_2,monomials), select, p -> degree(p)=m), _plus@op);
F := factor(x*fm[2]-y*fm[1])
```

$$3$$

$$\begin{pmatrix} x\,y^2 \\ -2\,y^3 \end{pmatrix}$$

$$-3\,x\,y^3$$

Wir führen also eine dritte x-Aufblasung durch und erhalten weiterhin einen stationären Punkt mit verschwindender Jacobi-Matrix.

```
f_3 := subs(normal(JI_x*subs(f_2,SPsi_x)), [xx=x,yy=y]); f_3 := map(normal(f_3/x^(m-1)),factor);
sol := solve(f_3,[x,y]); sol := sol[1]:
print(sol,subs(jacobian(f_3,[x,y]),sol));
```

$$\begin{pmatrix} x\,(x^3+x^2\,y^2) \\ x^5\,y^3+x^3\,y^3-3\,x^3\,y-3\,x^2\,y^3 \end{pmatrix}$$

$$\begin{pmatrix} x\,(y^2+x) \\ y\,(x^3\,y^2+x\,y^2-3\,x-3\,y^2) \end{pmatrix}$$

$$\left\{[x=0,\,y=0],\,\left[x=-\mathrm{i},\,y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right],\,\left[x=-\mathrm{i},\,y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\,\left[x=\mathrm{i},\,y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,\mathrm{i}\right)\right],\right.$$
$$\left.\left[x=\mathrm{i},\,y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,\mathrm{i}\right)\right]\right\}$$

$$[x=0,\,y=0],\,\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Jetzt verändert sich das charakteristische Polynom.

```
m := min(op(map(f_3,ldegree)));
fm := map(map(map(f_3,monomials), select, p -> degree(p)=m), _plus@op);
F := factor(x*fm[2]-y*fm[1])
```

$$2$$

$$\begin{pmatrix} x^2 \\ -3\,x\,y \end{pmatrix}$$

$$-4\,x^2\,y$$

Wir müssen nun parallel eine x- und eine y-Aufblasung durchführen, da x=0 jetzt eine mehrfache charakteristische Richtung ist. Wir beginnen mit der x-Aufblasung, die zu einem Sattelpunkt führt.

```
f_4x := subs(normal(JI_x*subs(f_3,SPsi_x)), [xx=x,yy=y]); f_4x := map(normal(f_4x/x^(m-1)),facto
sol := solve(f_4x,[x,y]); sol := sol[1]:
print(sol,subs(jacobian(f_4x,[x,y]),sol));
```

$$\begin{pmatrix} x\,(x^2\,y^2+x) \\ x^5\,y^3+x^3\,y^3-4\,x^2\,y^3-4\,x\,y \end{pmatrix}$$

$$\begin{pmatrix} x\,(x\,y^2+1) \\ y\,(x^4\,y^2+x^2\,y^2-4\,x\,y^2-4) \end{pmatrix}$$

$$\left\{[x=0,\,y=0],\,\left[x=-i,\,y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,i\right)\right],\,\left[x=-i,\,y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,i\right)\right],\,\left[x=i,\,y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,i\right)\right],\right.$$

$$\left.\left[x=i,\,y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,i\right)\right]\right\}$$

$$[x=0,\,y=0],\,\begin{pmatrix} 1 & 0 \\ 0 & -4 \end{pmatrix}$$

Nun führen wir die y-Aufblasung durch. Sie führt erneut zu einem stationären Punkt mit verschwindender Jacobimatrix.

```
Psi_y := [ xx*yy, yy ];
SPsi_y := [ x=xx*yy, y=yy ]:
J_y := jacobian(Psi_y, [xx,yy]):
JI_y := simplify(J_y^(-1)):
```

$$[xx\,yy,\,yy]$$

```
f_4y := subs(normal(JI_y*subs(f_3,SPsi_y)), [xx=x,yy=y]); f_4y := map(normal(f_4y/y^(m-1)),facto
sol := solve(f_4y,[x,y]); sol := sol[5]:
print(sol,subs(jacobian(f_4y,[x,y]),sol));
```

$$\begin{pmatrix} -x^4\,y^5-x^2\,y^3+4\,x^2\,y+4\,x\,y^2 \\ y\,(x^3\,y^5+x\,y^3-3\,x\,y-3\,y^2) \end{pmatrix}$$

$$\begin{pmatrix} -x\,(x^3\,y^4+x\,y^2-4\,x-4\,y) \\ y\,(x^3\,y^4+x\,y^2-3\,x-3\,y) \end{pmatrix}$$

$$\{[x=\sigma_1,\,y=\sigma_4],\,[x=\sigma_2,\,y=\sigma_3],\,[x=\sigma_3,\,y=\sigma_2],\,[x=\sigma_4,\,y=\sigma_1],\,[x=0,\,y=0]\}$$

where

$$\sigma_1=\sqrt{2}\left(-\tfrac{1}{2}-\tfrac{1}{2}\,i\right)$$

$$\sigma_2=\sqrt{2}\left(-\tfrac{1}{2}+\tfrac{1}{2}\,i\right)$$

$$\sigma_3=\sqrt{2}\left(\tfrac{1}{2}-\tfrac{1}{2}\,i\right)$$

$$\sigma_4=\sqrt{2}\left(\tfrac{1}{2}+\tfrac{1}{2}\,i\right)$$

$$[x=0,\,y=0],\,\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Das charakteristische Polynom zerfällt in drei Linearfaktoren.

```
m := min(op(map(f_4y,ldegree)));
fm := map(map(map(f_4y,monomials), select, p -> degree(p)=m), _plus@op);
F := factor(x*fm[2]-y*fm[1])
```

$$2$$

$$\begin{pmatrix} 4\,x^2+4\,y\,x \\ -3\,y^2-3\,x\,y \end{pmatrix}$$

$$-7\,x\,y\,(x+y)$$

Wir führen erneut eine y-Aufblasung durch und finden zwei stationäre Punkte. Der erste ist ein Sattelpunkt, der zweite ist nicht hyperbolisch. Mittels Normalformberechnungen kann man zeigen, daß der zweite stationäre Punkt ein stabiler Knoten ist.

```
f_5y := subs(normal(JI_y*subs(f_4y,SPsi_y)), [xx=x,yy=y]); f_5y := map(normal(f_5y/y^(m-1)),fact
sol := solve(f_5y,[x,y]); sol1 := sol[1]: sol2 := sol[6]:
J1 := subs(jacobian(f_5y,[x,y]),sol1): print(sol1,J1); linalg::eigenvectors(J1);
J2 := subs(jacobian(f_5y,[x,y]),sol2):print(sol2,J2); linalg::eigenvectors(J2);
```

$$\begin{pmatrix} -2x^4y^7 - 2x^2y^3 + 7x^2y + 7xy \\ -y\left(-x^3y^7 - xy^3 + 3xy + 3y\right) \end{pmatrix}$$

$$\begin{pmatrix} -x\left(2x^3y^6 + 2xy^2 - 7x - 7\right) \\ y\left(x^3y^6 + xy^2 - 3x - 3\right) \end{pmatrix}$$

$$\left\{[x=0,\ y=0],\ \left[x=-1,\ y=\sqrt{2}\left(-\frac{1}{2}-\frac{1}{2}\,i\right)\right],\ \left[x=-1,\ y=\sqrt{2}\left(-\frac{1}{2}+\frac{1}{2}\,i\right)\right],\ \left[x=-1,\ y=\sqrt{2}\left(\frac{1}{2}-\frac{1}{2}\,i\right)\right],\ \left[x=-1,\ y=\sqrt{2}\left(\frac{1}{2}+\frac{1}{2}\,i\right)\right],\ [x=-1,\ y=0]\right\}$$

$$[x=0,\ y=0],\ \begin{pmatrix} 7 & 0 \\ 0 & -3 \end{pmatrix}$$

$$\left[\left[-3,\ 1,\ \left[\begin{pmatrix} 0 \\ 1 \end{pmatrix}\right]\right],\ \left[7,\ 1,\ \left[\begin{pmatrix} 1 \\ 0 \end{pmatrix}\right]\right]\right]$$

$$[x=-1,\ y=0],\ \begin{pmatrix} -7 & 0 \\ 0 & 0 \end{pmatrix}$$

$$\left[\left[0,\ 1,\ \left[\begin{pmatrix} 0 \\ 1 \end{pmatrix}\right]\right],\ \left[-7,\ 1,\ \left[\begin{pmatrix} 1 \\ 0 \end{pmatrix}\right]\right]\right]$$

Wir zeichnen nun "rückwärts" die Phasenporträts der verschiedenen aufgeblasenen Systeme. Bei f_5y liegt ein Sattel und ein stabiler Knoten vor. Der Sattel führt zu einer Separatrix; der exzeptionelle Divisor zu einer weiteren.

```
a := -1.5: b := 1.5:
vf := plot::VectorField2d(f_5y, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_5y, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq1 := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
eq2 := plot::Point2d(-1,0, Color=RGB::Red, PointSize=2):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq1,eq2,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



Aus den beiden Stationären Punkten werden bei der Rücktransformation zu f_4y zwei (lineare Approximationen von) Separatrizen, deren Steigungen durch die x-Koordinaten der Punkte bestimmt sind: x=0, x=-y. Als weitere Separatrix finden wir den exzeptionellen Divisor y=0. Wir haben also jetzt einen stationären Punkt mit sechs Sektoren.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_4y, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_4y, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
```

```
sep := plot::Implicit2d(y+x, x=a..b, y=a..b, Color=RGB::Red):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



Das System f_4x besitzt einen einzelnen Sattelpunkt, der neben dem exzeptionellen Divisor x=0 noch zu der Separatrix y=0 führt.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_4x, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_4x, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



Für das System f_3 müssen wir Informationen von beiden Aufblasungen verarbeiten. Die y-Aufblasung führt zu den Separatrizen x=0 und y^2=-x; die x-Aufblasung zu der Separatrix y=0.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_3, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_3, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sep := plot::Implicit2d(y^2+x, x=a..b, y=a..b, Color=RGB::Red):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



f_3 entstand durch eine x-Aufblasung, deren exzeptioneller Divisor die y-Achse ist. Aus der quadratische Kurve y^2=-x wird die kubische Kurve y^2=-x^3. Ferner haben wir weiterhin die Separatrix y=0, so daß weiterhin drei Separatrizen vorliegen.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_2, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_2, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sep := plot::Implicit2d(y^2+x^3, x=a..b, y=a..b, Color=RGB::Red):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



Das Spiel wiederholt sich: Die Achsen bleiben als Separatrizen (x=0 verschwindet und erscheint wieder als exzeptioneller Divisor). Die Kurve hat nun Grad 5: y^2=-x^5.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_1, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_1, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sep := plot::Implicit2d(y^2+x^5, x=a..b, y=a..b, Color=RGB::Red):
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
sepy := plot::Line2d([0,a],[0,b], Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,sepx,sepy,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```



Bei der letzten Rücktransformation verschwindet der exzeptionelle Divisor x=0 und es bleiben nur noch zwei Separatrizen: y=0 und y^2=-x^7. Es verbleiben also vier Sektoren.

```
a := -2: b := 2:
vf := plot::VectorField2d(f_0, x=a..b, y=a..b, Color=RGB::Black):
pp := plot::Streamlines2d(f_0, x=a..b, y=a..b, MinimumDistance=0.1,
                          Color=RGB::Blue):
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sep := plot::Implicit2d(y^2+x^7, x=a..b, y=a..b, Color=RGB::Red):
```

```
sepx := plot::Line2d([a,0],[b,0], Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,sepx,
     ViewingBox=[a..b,a..b], Scaling=Constrained, TicksNumber=None,
     Width=120,Height=120)
```